
Change 1 

	Function name
	fl_IMS_Server_RemoveSecurityContext

	Reason for change
	In case of p_WaitForCNF set to false (SecurityRelease and RegInfoRelease) execution of procedure of IPSec release started while UE can be in still in process of closing of TCP connection.

We suggest to add 2 seconds delay to secure TCP connection is correctly closed on UE side before starting procedure of IPSec release.

	Summary of change
	Added 2 seconds delay before function fl_IPsec_Release

	TTCN module
	\Common\IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	Accepted in principle: The 2s delay may solve issues with test cases where the next step at the IMS PTC is initiated by the SS, but it will cause issues for test cases where the next step is initiated by the UE.

( We need to wait non-blocking for confirmation of closing the TCP connections before releasing the security context (see below)


Before change

    function fl_IMS_Server_RemoveSecurityContext(inout IMS_Server_Type p_ImsServer,

                                               IMS_RemoveSecurityContext_Type p_AllOrOldestOnly := oldestOnly,

                                               boolean p_WaitForCNF := true) runs on IP_PTC

  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    /* @sic R5-145732: support of 2nd security context sic@ */

    /* @sic R5s160735: p_AllOrOldestOnly sic@ */

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    while (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) { /* @sic R5s160735: "while" sic@ */

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

      v_SecurityInfo := p_ImsServer.ProtectedContext1.SecurityInfo.Protected;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      // Close TCP connection

      fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1, p_WaitForCNF);

      if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {   // if there is no other security context ...

        // ... stop TCP and UDP server

        f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

        f_TcpServer_Stop(v_NW_Address, v_LocalPortServer, p_WaitForCNF);

        p_ImsServer.ProtectedContext1 := omit;

      }

      else {

        p_ImsServer.ProtectedContext1 := p_ImsServer.ProtectedContext2;

        p_ImsServer.ProtectedContext2 := omit;

      }

      // Release security association

      fl_IPsec_Release(v_NW_Address, v_UE_Address, v_SecurityInfo);

      if (p_AllOrOldestOnly == oldestOnly) { /* @sic R5s160735 sic@ */

        break;

      }

    }

  }
After change

 function fl_IMS_Server_RemoveSecurityContext(inout IMS_Server_Type p_ImsServer,

                                               IMS_RemoveSecurityContext_Type p_AllOrOldestOnly := oldestOnly,

                                               boolean p_WaitForCNF := true) runs on IP_PTC

  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    /* @sic R5-145732: support of 2nd security context sic@ */

    /* @sic R5s160735: p_AllOrOldestOnly sic@ */

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    while (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) { /* @sic R5s160735: "while" sic@ */

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

      v_SecurityInfo := p_ImsServer.ProtectedContext1.SecurityInfo.Protected;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      // Close TCP connection

      fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1, p_WaitForCNF);

      if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {   // if there is no other security context ...

        // ... stop TCP and UDP server

        f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

        f_TcpServer_Stop(v_NW_Address, v_LocalPortServer, p_WaitForCNF);

        p_ImsServer.ProtectedContext1 := omit;

      }

      else {

        p_ImsServer.ProtectedContext1 := p_ImsServer.ProtectedContext2;

        p_ImsServer.ProtectedContext2 := omit;

      }

      f_Delay(2.0);        
      // Release security association

      fl_IPsec_Release(v_NW_Address, v_UE_Address, v_SecurityInfo);

      if (p_AllOrOldestOnly == oldestOnly) { /* @sic R5s160735 sic@ */

        break;

      }

    }

  }…
MCC160 Implementation
Enhancement of IMS_Server_Type

	  type record of IMS_SecurityContext_Type IMS_SecurityContextList_Type;

	type record IMS_Server_Type {                                 /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS, UTRAN) */

    IMS_SecurityContext_Type            UnprotectedContext,

    IMS_SecurityContext_Type            ProtectedContext1   optional,

    IMS_SecurityContext_Type            ProtectedContext2   optional,

    IP_AddrInfo_Type                    IPv4Addr            optional,   /* the server's IPv4 address, omit as long as the server is not started */

    IP_AddrInfo_Type                    IPv6Addr            optional,   /* the server's IPv6 address, omit as long as the server is not started */

    IPsec_SecurityKeys_Type             SecurityKeys        optional,   /* security keys to be applied for security associations */

    IMS_RegistrationAddress_Type        RegistrationAddress optional,   /* if present => UE has at least started the registration and in case of non-GIBA

                                                                           IPsec and UDP server for protected signalling are started */

    PDN_Index_Type                      PdnIndex            optional,   /* omit as long as the server is not started

                                                                           @sic R5s150648 Change 2.2: DrbInfo replaced by PdnIndex sic@ */
    IMS_SecurityContextList_Type        PendingSecurityContextList
  };


Change of function for closing TCP connections
	  function f_TcpServer_CloseConnection(template (omit) IP_Connection_Type p_IP_Connection,

                                       boolean p_WaitForCNF := true) runs on IP_PTC return boolean
  {

    var IP_Connection_Type v_IP_Connection;

    var boolean v_ConnectionIsClosed := true;
    if (isvalue(p_IP_Connection)) {

      v_IP_Connection := valueof(p_IP_Connection);

      fl_TCP_Close(v_IP_Connection, p_WaitForCNF);

      /* routing table does not need to be modified as the server is still running */
      v_ConnectionIsClosed := p_WaitForCNF;
    }
    return v_ConnectionIsClosed;
  }

	  function f_TcpClient_CloseConnection(template (omit) IP_Connection_Type p_IP_Connection,

                                       boolean p_WaitForCNF := true) runs on IP_PTC return boolean
  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    var IP_Connection_Type v_IP_Connection;

    var boolean v_ConnectionIsClosed := true;
    if (isvalue(p_IP_Connection)) {

      v_IP_Connection := valueof(p_IP_Connection);

      fl_TCP_Close(v_IP_Connection, p_WaitForCNF);

      // Modify routing table

      f_IP_PTC_RoutingTable_RemoveEntry(v_IP_Connection);

      v_ConnectionIsClosed := p_WaitForCNF;

    }

    return v_ConnectionIsClosed;
  }


Handling of pending connection
	  function fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(inout IMS_Server_Type p_ImsServer,

                                                                  template (omit) IMS_SecurityContext_Type p_SecurityContext) runs on IP_PTC

  {

    var IMS_SecurityContext_Type v_PendingSecurityContext;

    var integer i;

    if (isvalue(p_SecurityContext)) {

      v_PendingSecurityContext := valueof(p_SecurityContext);

      if (ispresent(v_PendingSecurityContext.TcpClient) or ispresent(v_PendingSecurityContext.TcpServer)) {

        // => wait for confirmation of TCP close

        i := lengthof(p_ImsServer.PendingSecurityContextList);

        p_ImsServer.PendingSecurityContextList[i] := v_PendingSecurityContext;

      } else {

        // Release security association

        fl_IPsec_Release(p_ImsServer.RegistrationAddress.NW, 

                         p_ImsServer.RegistrationAddress.UE, 

                         v_PendingSecurityContext.SecurityInfo.Protected);

      }

    }

  }

	  function fl_IMS_Server_SecurityContextCloseTCP(inout IMS_Server_Type p_ImsServer,

                                                 IMS_SecurityContextEnum_Type p_SecurityContextEnum,

                                                 boolean p_WaitForCNF := true) runs on IP_PTC return template (omit) IMS_SecurityContext_Type
  {

    var template (omit) IMS_SecurityContext_Type v_SecurityContext := fl_IMS_Server_GetSecurityContext(p_ImsServer, p_SecurityContextEnum);


    var IMS_SecurityContext_Type v_PendingSecurityContext;
    if (isvalue(v_SecurityContext)) {




      v_PendingSecurityContext := valueof(v_SecurityContext);

      if (f_TcpClient_CloseConnection(v_PendingSecurityContext.TcpClient, p_WaitForCNF)) {  // f_TcpClient_CloseConnection returns true if connection is closed 

        v_PendingSecurityContext.TcpClient := omit;

      }

      if (not f_TcpServer_CloseConnection(v_PendingSecurityContext.TcpServer, p_WaitForCNF)) {  // f_TcpServer_CloseConnection returns true if connection is closed 

        v_PendingSecurityContext.TcpServer := omit;

      }
      v_SecurityContext.TcpClient := omit;

      v_SecurityContext.TcpServer := omit;

      fl_IMS_Server_SetSecurityContext(p_ImsServer, p_SecurityContextEnum, v_SecurityContext); // @sic R5s150340: v_SecurityContext instead of v_SecurityContextValue, as v_SecurityContext has been changed sic@

      return v_PendingSecurityContext;
    }
    return omit;
  }

	  function fl_IMS_Server_RemoveSecurityContext(inout IMS_Server_Type p_ImsServer,

                                               IMS_RemoveSecurityContext_Type p_AllOrOldestOnly := oldestOnly,

                                               boolean p_WaitForCNF := true) runs on IP_PTC

  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    /* @sic R5-145732: support of 2nd security context sic@ */

    /* @sic R5s160735: p_AllOrOldestOnly sic@ */

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    var template (omit) IMS_SecurityContext_Type v_PendingSecurityContext;
    while (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) { /* @sic R5s160735: "while" sic@ */

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

      v_SecurityInfo := p_ImsServer.ProtectedContext1.SecurityInfo.Protected;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      // Close TCP connection

      v_PendingSecurityContext := fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1, p_WaitForCNF);

      if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {   // if there is no other security context ...

        // ... stop TCP and UDP server

        f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

        f_TcpServer_Stop(v_NW_Address, v_LocalPortServer, p_WaitForCNF);

        p_ImsServer.ProtectedContext1 := omit;

      }

      else {

        p_ImsServer.ProtectedContext1 := p_ImsServer.ProtectedContext2;

        p_ImsServer.ProtectedContext2 := omit;

      }

 

      fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(p_ImsServer, v_PendingSecurityContext);
      if (p_AllOrOldestOnly == oldestOnly) { /* @sic R5s160735 sic@ */

        break;

      }

    }

  }

	  template (value) IMS_Server_Type cs_IMS_Server_Init :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS, UTRAN) */

    UnprotectedContext := cs_IMS_SecurityContext(cs_IMS_SecurityContextInfo_Unprotected),  /* @sic R5-145732: support of 2nd security context sic@ */

    ProtectedContext1 := omit,                                                             /* @sic R5-145732: support of 2nd security context sic@ */

    ProtectedContext2 := omit,                                                             /* @sic R5-145732: support of 2nd security context sic@ */

    IPv4Addr := omit,

    IPv6Addr := omit,

    SecurityKeys := omit,

    RegistrationAddress := omit,                                                           /* @sic R5-145732: support of 2nd security context sic@ */

    PdnIndex := omit,                                                                      /* @sic R5s150648 Change 3 sic@ */
    PendingSecurityContextList := {}
  };


Handling of TCP CLOSE CNF
	  altstep a_IMS_TCP_Close_CNF_Handler(inout IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var IP_SOCKET_IND v_TCP_CLOSE_CNF;

    var IP_Connection_Type v_IP_Connection;

    var IMS_SecurityContextList_Type v_SecurityContextList;

    var IMS_SecurityContext_Type v_SecurityContext;

    var integer i;

    var integer v_Length;

    [] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(?)) -> value v_TCP_CLOSE_CNF {

      v_SecurityContextList := p_ImsServer.PendingSecurityContextList;

      v_Length := lengthof(v_SecurityContextList);

      if (v_Length > 0) {

        p_ImsServer.PendingSecurityContextList := {};      // reset list of pending contexts

        v_IP_Connection := v_TCP_CLOSE_CNF.CTRL.ConnectionId;

        for (i := 0; i < v_Length; i := i + 1) {

          v_SecurityContext := v_SecurityContextList[i];

          if (match(v_SecurityContext.TcpClient, v_IP_Connection)) {

            v_SecurityContext.TcpClient := omit;

          } else if (match(v_SecurityContext.TcpServer, v_IP_Connection)) {

            v_SecurityContext.TcpServer := omit;

          }

          fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(p_ImsServer, v_SecurityContext);   // Either relase IPsec or add security context to list of pending contexts

        }

      }

    }

  }

	  altstep a_IP_IMS_TCP_ConnectionControl(inout IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var boolean v_ImsServer_IsStarted := f_ImsServer_IsStarted(p_ImsServer);

    var boolean v_SecurityContext1_IsStarted := v_ImsServer_IsStarted and fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1);

    var boolean v_SecurityContext2_IsStarted := v_ImsServer_IsStarted and fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2);

    [] a_IMS_TCP_Close_CNF_Handler(p_ImsServer);
    [v_ImsServer_IsStarted] a_IMS_Server_SecurityContextAcceptTCP(p_ImsServer, unprotected);

    [v_SecurityContext1_IsStarted] a_IMS_Server_SecurityContextAcceptTCP(p_ImsServer, protectedContext1);

    [v_SecurityContext2_IsStarted] a_IMS_Server_SecurityContextAcceptTCP(p_ImsServer, protectedContext2);

    [v_ImsServer_IsStarted] a_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

    [v_SecurityContext1_IsStarted] a_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

    [v_SecurityContext2_IsStarted] a_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

  }

	  altstep a_TcpServer_AcceptConnection(out template (omit) IP_Connection_Type p_IP_Connection,

                                       template (present) IP_Socket_Type p_ServerSocket,

                                       template (present) IP_Socket_Type p_RemoteSocket := cr_IP_Socket_Any) runs on IP_PTC

  { 
    var IP_SOCKET_IND v_IP_SOCKET_IND;

    [] IP_SOCK_CTRL.receive(cr_TCP_ACCEPT_IND(p_ServerSocket, p_RemoteSocket)) -> value v_IP_SOCKET_IND {

      /* @sic R5-145732: new connection is checked and handled by the caller sic@ */

      p_IP_Connection := v_IP_SOCKET_IND.CTRL.ConnectionId;

      /* the routing table does not need to be changed as there is an entry for the TCP server already and there is no need to change it just because of the remote address */

    }





  }


